
A local area network in its simplest terms is a group of personal computers wired to central serving 

unit that provides file handling, printing resources, communications devices, and other hardware and 

software. No matter what configuration of network you have (token ring, STARLAN, Etherlink, 

ARCnet, and so on), some basic principles apply. 

Workstations (usually PCs) are computers or terminals connected by cable or fiber optics to a file 

server and in some cases to multiple servers that control printing communications or other network 

duties. The file server is large central storage unit for all of the workstations connected to it. 

Whenever a workstation requests a service, the file server performs all of the disk operations 

necessary to supply the workstation with data and program code. 

All of the actual computation is done locally at the workstations. The file server simply passes 

information in data packets that are addressed to a particular workstation much like a letter might be 

addressed to a particular mailbox. What the workstation does with data before writing it back to the 

file server’s hard disk is of no real concern to the network operating system. It is simply information 

to be transferred. 

The hard drive on a file server is usually quite large. The hard disk is arranged in subdirectories that 

can be accessed by any workstation that has been given clearance. By utilizing user names and 

passwords, the network operating system provides a means of logging on to the network. A user who 

has a valid password is given a predetermined level of access to the network services. This access 

also restricts the user to using files in only those subdirectories he or she has permission to use. The 

System Administrator decides what files a particular user may alter by setting some files as “read-

only.” 

 

Programming for Multiple Users 

Typically a user Iogs onto a network with a user name and a password. This combination is 

recognized by the network operating system as representing a user who has certain rights within the 

system. These rights include what subdirectories the user can access; files the user may create, erase, 

or alter; where the user can send print jobs; and so on. If more than one user is given rights to a 

particular directory, it is possible for multiple users to access the same files at the same time - a 

potentially dangerous situation for data. 

For example, suppose that two users of the network simultaneously open the customer data file and 

both happen to be looking for John Smith. Each user gets a copy of the John Smith record, and each 

of them makes a different change to the record. User A is a little faster than User B, so she sends her 

changes back to the file server first, and the changes she has made become part of the .DBF file. User 

B, who is a little s to record changes, then writes his changes back to the file server. In this case the 

record as User B has defined it becomes the content of the .DBF file. All of the changes made by User 

A are eliminated because User B’s file over writes User A’s file. And anything not changed by User B 

remains the same as the old contents. The data has now been corrupted. 

Another scenario is that User A and User B have exactly the same typing speed, and both try to write 

their amended records to the data file at exactly the same time. This data collision could cause even 

more damage to the condition of the data file, possibly scrambling it so badly that it would become 

unusable. 

In any situation where two users might try to perform database functions on the same file at the 

same time, file and record locking procedures are necessary. 

The idea of writing multiuser applications is scary enough to turn off some programmers. Clipper, 

however, makes it nearly painless. Remember that you seldom have to change what the program 

does in order to make it network compatible. You need only to make certain that the environment is 

conditioned properly before doing reads and writes to files. 

There are three main concepts to understand before you write multiuser applications: file usages. file 

locks, and record locks. The capability to perform these three tasks makes multiuser applications a 

reality. 

File Usage — Exclusive or Shared 

When you write the code for a program that opens a file, it is necessary to determine whether this 

file can be shared by other operators. If all the operators are going to make changes to only one 

record at a time, there is no reason not to let the file be shared. If one of the users uses the PACK 

command on the file and then re-creates index files, it is not possible for more than one person to 

have access to the file. 

Use the SET EXCLUSIVE OFF  command to provide shared access. By default a Clipper application 

opens all files in an EXCLUSIVE manner, which means no other user may also open it. 

File Locking 

A file lock grants the operator temporary, EXCLUSIVE use of the file. While the file lock is on, other 

users cannot make any changes to the data; they are locked out of making writes to the file. A file 

lock, however, does nothing to prevent other users from reading records in the file. 

Record Locking 

A record lock provides an operator with temporary, EXCLUSIVE use of just one record in the file. 

While the record is locked, other users may read it, but no other user may write to it. 

 

Network Compatibility of Clipper 

One of the most attractive features of Clipper is that networking functions are built in. There is 

nothing extra to supply to a user, and there are no royalty or license fees to pay with each copy that 

is distributed. You create the application, you own it, and you give it to the users that you want to 

have it. 

With the proper network configuration and a Clipper application written with multiuser situations 

and network security in mind, operators can perform day-to-day tasks with the application without 

giving much thought to the fact that they are sharing resources with other users. A program designed 

for data entry can be running simultaneously many times on the network so that several operators 

can be using the same application to enter records into the same database at the same time. 

All of the file and record handling done in a Clipper application strictly adheres to standards 

introduced with PC DOS and MS-DOS® V3. 1. Clipper does not make any nonstandard hardware calls. 

All of Clipper’s file and record locking activities are done with calls to MS-DOS. Any network that is 

compatible with PC/MS DOS V3.1 file and record locking calls -and most are- provides the proper 

environment for running your Clipper applications. To maintain full compatibility with Clipper, all of 

your work stations must be running under MS-DOS V3.1 or greater. 

You should be familiar with all the requirements of your network operating system. Each brand of 

local area network software has its own small quirks and idiosyncrasies. 

Single-User Application on a Network 

The only time you need to be concerned about file and record locking in a Clipper application is when 

multiple users will be sharing the same data and index files. A Clipper application that does not allow 

multiple users access to the same data files does not need to include any file or record locking 

precautions even if the applications runs on a network. Because it is so easy to place file and record 

locking capabilities into an application, however, you may decide to write the application to be 

multiuser compatible just to save time later when you are asked to do so.  

Controlling Clipper in a Network Environment 

Only a small number of the Clipper commands and functions come into play when you move into a 

networking environment. 

The SET EXCLUSIVE ON/off Command 

The SET EXCLUSIVE command in a Clipper application determines whether files are opened in 

shared mode or in EXCLUSIVE mode. By default Clipper USEs files in EXCLUSIVE mode. When you 

enter SET EXCLUSIVE OFF , Clipper expects the rest of the application to adhere to multiuser 

conventions. 

The USE Command 

One of the options of the USE command is EXCLUSIVE. When you include USE for a file in 

EXCLUSIVE mode, you create the ultimate file lock because no other user can even open the file in 

a read-only situation, 

The APPEND BLANK Command 

In a networking environment, the APPEND BLANK command creates a new record at the end of the 

file and places an RLOCK()  on the new record in one step. 

The FLOCK() Function 

The FLOCK()  function enables you to have temporary EXCLUSIVE use of a file that is already open 

in shared mode. During the time FLOCK()  is active, no other users can make changes to any of the 

records. Other users may, however, still read the contents of the file.  

The RLOCK0 Function 

The RLOCK()  function requests temporary EXCLUSIVE use of an individual record. While the RLOCK()   

is active, other users may read but not write to that record. 

The UNLOCK Command 

File and record locks applied at a particular workstation can be canceled only at that same 

workstation. A network user at one workstation cannot break file and record locks established at 

another workstation. The UNLOCK command is used within an application to release locks that have 

been applied so that other users may have access. 

 

The NETERR() Function 

The NETERR() function returns a logical value whenever you try to USE a file or APPEND BLANK. The 

default value of NETERR() is false (.F.). 

NETERR () returns true (.T.) under the following conditions: 

� You try to USE a file that is currently being USEd EXCLUSIVEly by another workstation. 

� You have tried to USE EXCLUSIVEIy a file that is already open on another workstation. 

� You have tried to APPEND BLANK in a file that is temporarily locked with the FLOCK() function. 

� You have tried to APPEND BLANK into a file at exactly the same time as another user. 

Developing Applications for Networks 

Clipper makes it easy to develop network applications -even on an independent (not networked) 

system. When you SET EXCLUSIVE OFF, Clipper expect all other commands and functions to operate 

within the multiuser environment. If you forget to lock a record before you try to write to it, Clipper 

will object and display an error message. This is extremely useful in debugging because these error 

messages will be returned even on a system that is not connected to a network. 

Try entering this: 

SET EXCLUSIVE 0FF 
USE datfile 
GOTO 12 
REPLACE strl WITH “string” 
WAIT “You will never see this message” 

 

What you will see is a DOS prompt and an error message telling you that there is a lock required. If 

you are using standard procedures and well tested user-defined functions, you do not have to 

develop network applications on a network. 

Programming with a Local Area Network in Mind 

Whenever you set out to create a multiuser application, you need to work out the details of multiple 

usage. A large part of your decision-making is determined by the type of application that you are 

building.  

� Are entries to the database short, or does the input screen have 65 entries?  

� Will certain data files in the application be used more often than others?  

� Will the System Administrator need to do massive maintenance -that is, bring in large 

numbers of records from a mainframe- several times a day?  

� Is the data the type of information that usually is only entered once, such as 

historical tracking or research compilation, and then updated only to correct typos?  

� Is the data fairly static, such as price lists and museum inventory, and referred to 

more than it is written?  

� Only you will be able to determine the exact file and record locking scheme that best 

suits your applications. 

 

Designing Your File-Handling Routines 

If you put a group of the country’s best Clipper programmers together with the technical support 

staff at Nantucket, you could spend the entire day discussing the philosophies and technologies of 

file and record locking and never come to more than a respectful agreement that, “Yeah, you could 

do that,” or, “lt’s doable.” 

There is no trick, however, to recognizing when to lock files and records: 

� If more than one person has access to the file, design the program to place a lock    on 

the file before writing to it. 

� Use a record lock  if the code alters only one record. 

� Use a file lock if the code will change more than one record, and the operation is short. 

� Use the file EXCLUSIVEly  if many commands will change more than one record. 

 

Determining Persistence in File and Record Locking 

The trick in designing file and record locking is in determine what in do if the lock   fails. When you 

design your file and record locking scheme, the main ingredient in consider is the concept of 

persistence.  

• How hard does your program try to implement the lock   ?  

• What does the program do when a lock   is rejected by the network operating system? 

There’s a wide latitude of persistence for lock design. You could decide simply in abort the module 

and return in the menu after telling the operator that the attempted action cannot be performed at 

the moment. Or if you know that other entries to the data file are going in be short, you may decide 

to attempt the lock repeatedly for a few seconds before telling the operator the procedure cannot be 

done. In addition, your program could try repeatedly to lock the record or file forever or until the 

request is granted. 

The decision is yours. The best advice is to start in the “middle of the road” and make changes as you 

become comfortable with the environment. Begin using a scheme that is between an infinite wait 

and no wait at all. Then, as you get in know your operators; you can make program adjustments that 

are the most comfortable. 

 

Two File and Record Locking Schemes 

The Clipper factory disks contain a file called LOCKS.PRG, a file and record locking scheme that 

depends on the programmer to set the length of persistence. LOCK S.PRG provides four user-defined 

functions to set the persistence level anywhere from zero in infinity. These functions can be used as 

the foundation for a file and record locking scheme. They compose basic, no-nonsense code that 

attempts in perform the action once each second until either the time you have inc as an argument is 

expired or the lock is achieved. 

The one area these functions do not address is the user interface that each application needs. The 

file and record locking system that is used for the invoicing application in Part II is based in part on 

the basic structures introduced in LOCKS.PRG. The invoicing application, however, provides a standard 

user interface enables the user in decide issues of persistence. 

Study the code in LOCKS.PRG to understand why the functions work. Next, look at the following user-

defined functions to see the basic differences in philosophy. Then design a file and record locking 

scheme to suit yourself or use as a plug-in module the set that appeals most to you. 

Opening Files on a Local Area Network 

The simplest of all ways to use a file in a network environment is simply to begin your application: 

USE datfile 

In order to USE files in shared mode within a Clipper app you must use the SET EXCLUSIVE OFF 

command. When you have done so, your application opens files in shared mode. 

Never open data and index files with the same command. Trying to USE an unavailable data file is a 

much lower-level error than trying to open index file s for which no data file is available.  

Trying to open an index file without a data file triggers a Clipper run-time error. This aborts your 

application. Attempting to USE a data file that is unavailable simply returns the NETERR() function 

as true ( .T.). 

The expression USE datfil INDEX dat1, dat2, dat3 is dangerous because if datfil  is not 

available, NETERR() will be set to true (.T.), and the command then will try to open the index files. The 

data file is not open, so you get the run-time error message. 

When you open a file, test to make sure that a network error has not occurred. When you have 

access to the data file, you can add index files, close index files, change their order, and so on 

without regard for consequence. Index file handling is totally transparent. 

The following is the simplest way to open a network file: 

SET EXCLUSIVE OFF 
USE filename 
IF .NOT. NETERR()  
    SET INDEX TO filename 
ELSE 
   ? “NETWORK ERROR HAS OCCURED, FILE NOT OPEN.” 
ENDIF 
 

Index files that are opened along with a shared data file automatically reflect any file on record 

locking situations applied to the data file. 

Opening a File with USE_UDF() 

Here you have the issue of persistence. What do you do if the USE is refused? This user-defined 

function leaves persistence for the operator to decide. The function tries to open the file for five 

seconds and then prompts the operator for a decision. 

*   USE_UDF function 
*   ATTEMPTS TO OPEN A FILE FOR NETWORK USE IN EITH ER 
*   SHARED OF EXCLUSIVE USE 
*   PARAMETERS USED: 
*       1. FILE NAME (BE SURE TO SURROUND IN OUOTES ) 
*       2. MODE ( .T. = EXCLUSIVE .F. = SHARED) 
*   Example usage: 
*   IF USE_UDF( “CUST”, .T. ) 
*      SET INDEX TO cust, lnam, bal 
*   ELSE 
*      CLEAR 
*      CLEAR ALL 
*      RETURN  
*  ENDIF 
 
FUNCTION USE_UDF  
PARAMETERS datfil, usemode  
delay = 5 
DO WHILE T. 
   IF usemode && Attempt to open exclusive 
      USE &datfil, use_mode    
   ELSE 
      USE &datfil     && Attempt to open shared mod e 
   ENDIF 
   IF .NOT. NETERR()  && File is opened successfull y 
      RETURN .T.      && Go back to calling procedu re 
                      && with the value of the func tion 
                      && returned as True 
   ELSE 
      IF delay > 0 
         INKEY( 1 ) 
         delay = delay — 1 
         LOOP 
       ELSE  
         SAVE SCREEN 
         conf = "  " 
         @  8, 15 clear TO 20, 60  
         @  8, 15 TO 20, 60 DOUBLE 
         @ 10, 20 SAY "Data File: " + datfil 
         @ 12, 20 SAY "This file is unavailable at this time."  
         @ 14, 20 SAY "Do you want to continue atte mpting"  
         @ 16, 20 SAY "to open it for another five seconds?" 
         @ 18, 20 SAY "                (Y/N) ------ ------>" 
         SET CURSOR OFF        && These functions d o 
         WAIT TO "" TO CONF  && not use a READ 
         SET CURSOR ON        && command, so they  
         IF UPPER(conf) = "Y" && can be included as  
            delay = 5        && part of a function 
            RESTORE SCREEN  && invoked with a 
            LOOP              && VALID clause in a 
         ELSE              && GET / READ construct 
            RESTORE SCREEN 
            RETURN .F. 
         ENDIF 
      ENDIF 
   ENDIF    
 

Look at the example usage. You will see the line : 
 
IF USE_DBF( "CUST", .T. ) 

The USE_UDF () function will  return a logical true (.T.) if the file is USEd successfully. If the USE_UDF() 

function returns .T., it is perfectly safe in open your index file. If the function returns false (.F.), you 

will need to handle the error. In this case, 

CLEAR 
CLEAR ALL 
RETURN 

 

is only an example, not a recommendation. 

You must pass two parameters in the USE_UDF()  function: datfil  and use_mode . The datfil 

parameter is used in hold the name of the file you wish to open, whereas the use_mode  parameter 

determines whether to try to open it exclusively. 

The function opens up a loop, which is much easier to deal with when it is a separate procedure and 

doesn’t complicate the main routine, and then tries to open the file in an exclusive or nonexclusive 

mode, depending on how you set the use_mode  parameter. 

If the file is opened successfully and a NETERR() does not occur, control is passed back to the 

calling module, the USE_UDF()  function has a value of true (.T.), and the user-defined function is 

terminated. 

If there is a NETERR(),  the user-defined function must now handle the failure in get the requested 

service. The command 

delay = 5 

determines the number of seconds during which the function attempts to get the file. You can adjust 

this by simply changing the delay value. Remember that making the delay too long may cause the 

operator to think that the program has bombed. Operators who think the software has bombed have 

the dangerous tendency in reboot computers and take other drastic actions. 

The USE_UDF()  function then tries to use that file one time per second for however many seconds 

you have set for the delay. You might also choose to change the line: 

delay = delay - 1  

to  

delay = delay - .5 

This would double the number of attempts within the delay frame. 

While the loop is within the delay frame, the USE_UDF function repeatedly tries to open the file. If, 

after the delay time has counted off the file is still not opened (NETERR() <> .T.), the USE_UDF() function 

will save the screen and display a message box. 

The message box serves two purposes. First, it tells the operator that the file is unavailable at the 

moment. Second, it places the persistence factor in the operator’s hands by asking whether the 

operator would like to continue waiting for that file. 



10 

 

If the value of conf  is Y, the function will reset the delay and continue the loop. If the operator 

chooses not to wait, the USE_UDF()  function will  be terminated and its value returned as false (.F.). 

If USE_UDF()  is false, the application cannot continue because in all likelihood the operations that 

are programmed into the module require the file in be open. Trying to perform operations on 

unavailable files creates run-time errors that abort the program. You need a way in back out of the 

trial gracefully so that the program can return you to the menus. See the invoicing program in Part II 

for some ideas. 

In reality, all you need to know in order to use this function is what you want the application to do if 

the USE command fails. 

Creating a New Record with APND() 

When the file and its associated index files are opened, you may proceed to use the file as you 

normally would. However, if you want in append a new record to the file, make sure that no one else 

has locked the file or is attempting to append a blank record at exact the same moment as you are. 

In the network environment, APPEND BLANK not only creates the new record, but locks it as well 

with the RLOCK() command. 

The safeguards for APPENDing new records are simple. APPEND BLANK is a command that executes 

quickly. Two simultaneous APPEND BLANK requests cause one of the requests to be rejected; but 

because the command doesn’t make long in complete, the second request is almost always honored. 

Persistence is seldom a problem here. The request normally is honored on the first or second try. The 

only other situation that might cause an APPEND BLANK in fail is when another workstation has 

temporarily placed a file lock. In such situations persistence comes into play. 

Here is the APND() function from the invoicing program: 

* APND function 
* 
* Attempts to append a blank record to the file and  lock it. 
* If the append is successful, the function returns  .T. 
FUNCTION APND 
delay = 5 
 
DO WHILE .T. 
   APPEND BLANK 
   IF NOT. NETERR()  
      RETURN .T. 
   ENDIF 
   IF delay 0 
      INKEY(1) 
      delay = delay — 1 
      LOOP 
   ELSE 
      SAVE SCREEN 
      conf = 
      @  8, 15 clear TO 18,60 
      @  8, 15 TO 18,60 DOUBLE 
      @ lO, 20 SAY "New record creation was unsucce ssful." 
      @ 12, 20 SAY "Do you want to continue attempt ing" 
      @ 14, 20 SAY "to add it for another five seco nds?" 
      @ 16, 20 SAY "     (Y/N) ——--—>" 
      SET CURSOR 0FF 

      WAIT "" TO CONF 
      SET CURSOR ON 
      IF UPPER(conf) = "Y"  
         delay = 5  
         RESTORE SCREEN  
         LOOP 
      ENDIF 
   ELSE  
      RESTORE SCREEN  
      RETURN .F. 
   ENDIF 
ENDD 
 
The APND()  function uses the exact same interface as USE_UDF()  to tell the user that a requested 

service was unsuccessful. The APND()  function does not require any argument because the APPEND 

BLANK command automatically creates the new record with an RLOCK()  command in place. Again, 

the delay is set for five seconds. 

Within a loop, the APPEND is attempted. If an error does not occur, the value of the function is set to 

true (.T.), and the APND() function is terminated. 

If a network error did occur, the user-defined function will attempt the APPEND BLANK command 

once each second for as many seconds as the delay is set. If the time expires without successful 

completion of the APPEND BLANK command, the user interface displays a message to the operator 

that the attempt to create the new record was unsuccessful. If the operator declines to wait for the 

APND() function to create the new record, your application must be able to back out gracefully. 

Locking the Current Record with RLOK() 

The RLOK()  function attempts to place a lock on the current record—that is the record in the 

current work area where the record pointer is set. If a record lock is successful, no other user may 

make any changes in this data until the lock has been released. 

Apply the record lock before the fields are put into variables for editing. This ensures that the 

operator is working with real data. A second operator who also wants to make changes to the record 

will be denied access. The second operator can only read the record after modifications become 

written to the file. This ensures that all information on the screen is the same as the contents of that 

record on disk. This concern for record locking prior to storing data in variables seldom applies to 

operator requests for data for read-only files. If your code does not let the user change a file, it is 

perfectly safe in let the user read it. 

Consider this scenario: User A and User B both read the data in the same record. User A immediately 

begins to make major changes. The data on User B’s screen is no longer up-to-date. User A finishes 

editing, and his Clipper application writes the record back to the file. User B, having been interrupted 

by a phone call, then decides that just one item needs to be changed. Because User B has an old copy 

of the record, when she writes the data back to the file, all of the changes made by User A are 

nullified. Placing the lock before data can be read is the best method for preventing this type of 

problem. 

This approach has its drawbacks as well. An operator might walk away from the computer for a white 

and leave the record locked for an excessive amount of time. This is a user training issue; operators 

must be taught multiuser etiquette. Fortunately, the situation does not cause data corruption. 

The RLOK () function does not require parameters to be passed to it. Here is the code: 

 
* RLOK function 
*  
* Attempts to lock the current record 
* for temporary exclusive use. 
* No parameters are needed to be passed. 
* 
* Example usage: 
* IF RLOK() 
* REPLACE FIELD1 WITH var1 
* REPLACE FIELD2 WITH var2 
* REPLACE FIELD3 WITH var3 
* ELSE 
* CLEAR 
* CLEAR ALL 
* RETURN 
* ENDIF  
FUNCTION RLOK  
delay = 5  
DO WHILE .T. 
   IF RLOCK()       && Attempt to lock the record. 
      RETURN (.T.) && If successful go back to call ing 
                        && procedure with the value  of the 
                        && function returned as Tru e. 
   ENDIF 
   IF delay > 0 
      INKEY(1) 
      delay = delay - 1 
      LOOP 
   ELSE 
      SAVE SCREEN 
      conf = "" 
      @  8, 15 clear TO 18,60 
      @  8, 15 DOUBLE  
      @ 10, 20 SAY "Record locking was unsuccessful ." 
      @ 12, 20 SAY "Do you want to continue attempt ing" 
      @ 14, 20 SAY "to lock it for another five sec onds?" 
      @ 16, 20 SAY "          (Y/N)-———>” 
      SET CURSOR 0FF 
      WAIT "" TO CONF 
      SET CURSOR ON 
      IF UPPER(conf) = "Y"  
         delay = 5  
         RESTORE SCREEN  
         LOOP 
      ELSE 
         RESTORE SCREEN 
         RETURN .F. 
      ENDIF 
   ENDIF 
ENDDO 

 

The RLOK()  function has a great deal in common with the other user defined functions presented in 

this chapter. The delay structure is almost exactly that of the others, and the user interface is the 

same except for the messages. 

Locking the Current File with FLOK() 

The FLOK()  user-defined function does exactly the same thing as the RLOK()  function except that 

the scope of the function covers the entire file rather than a single record. The two functions work 

the same way. 

* FLOK function 
* 
* Attempts to lock the current file 
* for temporary exclusive use. 
* No parameters are needed to be passed. 
*  
* Example usage: 
* IF FLOK() 
*    AVERAGE bal TO avbal FOR bal > 5000 
* ELSE 
*    CLEAR 
*    CLEAR ALL 
*    RETURN 
* ENDIF  
 
FUNCTION FLOK 
delay = 5 
DO WHILE .T. 
   IF FLOCK()       && Attempt to lock the file. 
      RETURN (.T.)  && If successful, 90 back to ca lling 
                    && procedure with the value of the 
                    && function returned as True. 
   ENDIF 
   IF delay > 0 
      INKEY(1) 
      delay = delay - 1 
      LOOP 
   ELSE 
      SAVE SCREEN  
      conf = "" 
      @  8, 15 clear TO 18, 60  
      @  8, 15 TO 18,60 DOUBLE 
      @ 10, 20 SAY "File locking was unsuccessful."   
      @ 12, 20 SAY "Do you want to continue attempt ing"  
      @ 14, 2O SAY "to lock it for another five sec onds?"  
      @ 16, 20 SAY "          (Y/N)----->" 
      SET CURSOR 0FF  
      WAIT "" TO CONF 
      SET CURSOR ON 
      IF UPPER(conf) = "Y" 
         delay = 5 
         RESTORE SCREEN  
         LOOP 
      ELSE 
         RESTORE SCREEN 
         RETURN .F. 
      ENDIF 
   ENDIF 
ENDDO 

Implementing Record and File  

Locking in Your Application 
 

Now that you have a basic set of tools to use, you can plug the tools into the proper places and get 

on with making the program do what you want it to do and with the assurance that multiple users 

can run it. 

Lock a record whenever the possibility exists for the flow of the program to make any change to that 

record. Lock the entire file when you perform commands that affect the contents of more than one 

record in the data file. For example: 

REPLACE ALL balance WITH balance * .05 FOR lastpay < CTOD ("01/01/89") 

Because the scope of this command probably alters more than one record within the file, the entire 

file should be locked. 

Read-only operations can be processed without regard to record locks. Any file or record can be read 

even though it is locked. That means a report can run while the data is changing. By the time the 

print job is completed, the data is no longer absolutely current. In many respects this is nothing new. 

Most reports are somewhat dated by the time they are actually used. 

A report usually is understood to be a snapshot of the data at a particular point in time. Once in a 

while, however, you need to know the exact total of items in the file.  Whether it is a report or one of 

the summary commands (AVERAGE, SUM, COUNT, TOTAL ), you may need to lock the file or USE 

it EXCLUSIVEly in order to get absolute numbers, in most cases, it is perfectly acceptable to let a 

report run through the record lock: the next report will reflect the new data. 

It is worth repeating that you should place the record lock before you allow the user to edit the data. 

Avoid letting the operator make 10 minutes worth of entries, only to inform him or her that you can’t 

get the record to lock. Whenever the operator chooses to edit data, apply the lock, and then fill 

variables with the values. Edit the variables, not the fields. If the operator wants to back out, all the 

program has to do is UNLOCK the record without performing any REPLACE commands. 

 

Files Other Than Data  

Files and Index Files 

 
Some provisions not expressly envisioned in the Clipper file and record locking schemes are MEM 

files, ALTERNATE files, and other files that you might be creating with the low-level F functions, such 

as FCREATE(), FSEEK(), FWRITE(),  and so on. 

A powerful method of handling these situations is to build a semaphore database, so named because 

it uses records as flags. Each type of operation that you wish to control is given one record in the 

.DBF file. Then all you have to do is to use the capability to RLOCK()  that record as a determinant. 

For example, when an app needs to create or update a .MEM file, it must first set the record pointer 

to the proper record in the semaphore file and lock it. Only one user can have a record locked at any 

given time, so this is an absolute indication that it is safe to manipulate the .MEM file. 

The invoicing application in Part II uses a semaphore database with just one record to protect the 

integrity of the next invoice number. A quick look at the source code reveals how simple it is to 

construct and use a semaphore database. 

A semaphore database introduces other possibilities, such as locking messages that tell one user 

what other user owns the lock and for what duration of time the lock has existed. A dynamic 

semaphore gives you the opportunity for including real-time updated messages. If your application 

were to write a record to a semaphore file stating the conditions of the file or record lock 

(EXCLUSIVE use, record numbers. time stamps. and so on), the file and locking functions can report 

more information to the operator whose request for service was denied. 

For example, when the application locks a record, the application writes a record to the semaphore 

file. The fields take entries for the file name, record number. TIME() , procedure, and the user’s 

name. Then, when the lock is released, the semaphore record is deleted. If another user requests a 

lock for the same file at the same record, the locking function attempts the lock. If the lock fails, the 

application then searches the semaphore database for a record that corresponds to the failed 

attempt.  In this case it probably is efficient to use the LOCATE command, because there would be 

only a small number of records (no more than one per user) to evaluate. 

For example: 

LOCATE FOR datfil = ‘cust” .AND. RECN() = 35 
IF .NOT. EOF() && or you could use IF FOUND() 
   @ 10, 10 SAY "User " + TRIM(user_nam) + " has th at record locked."  
   @ 11, 1O SAY "The lock was placed at:" + lok_tim e 
ENDIF 
 

File-Handling Commands and Functions 

A number of Clipper file manipulation commands require exclusive use of the data file that is 

affected by the command. The following is a list of those commands: 

� COPY STRUCTURE TO file 
� COPY TO file 
� CREATE file 
� INDEX ON TO file 
� JOIN ... TO file 
� SORT TO file 
� REINDEX 
� PACK 
� ZAP 

A number of commands requires that the file or the record be locked or the file be used exclusively.  

These commands include the following:  

� @ ... SAY … GET 
� APPEND FROM 
� DELETE 
� RECALL 
� REPLACE 
� UPDATE ON 

 
The first command applies only to those commands that GET. Some commands can affect more than 

one record, depending on the scope of that particular use of the command. For example: 

REPLACE balance with 500 && Affects only the curren t 
                                                                                     && record. A reco rd lock is                                                                          
                                && adequate. 

 
* This REPLACE has the potential to affect more tha n one 
* record. Either USE EXCLUSIVE or use a file lock. 
REPLACE ALL balance WITH balance * .015 ; 
                    FOR inv_dat < CTOD() 

 
Summary 

After reading this chapter, you should have a good idea of the issues involved with letting multiple 

users have access to data files simultaneously. 

You also should have a set of tools to accomplish the task. The record and file locking functions 

included in this chapter make up a complete set that you can use verbatim. 

 

Quoted from : Using Clipper (Que) by W.Edward Tiley 

 


